/*The SAS program (HEI-2010 Individual Scores by person using multiple days of NHANES 2003-2004 data (and MPED))

ByPerson.SAS*/

/*INSTRUCTIONS – complete tasks 1-4 in this section, and run these SAS codes before proceeding to the HEI-2010 scoring program that follows*/

/*1. Create a folder on your computer “home folder”, and save the FPED data, NHANES data, Demographic data, and the required HEI-2010 macros in it. Specify the path to the folder. */

%let home = C:\Users\Documents\FPED_NHANES; /*In this Example, the “home” folder is in C Drive, within Documents, and is called FPED_NHANES. */
/*2. Libnames here specify the input files. */

libname NH “&home\NH”; 
libname MPED “&home\MPED”; /*In this Example, the MPED data are in a folder called “MPED”, and the NHANES and Demographic data are in a folder called “NH”, all saved within the “home” folder. These are SAS datasets. */
/*3. Create a folder in the "Home" folder, where the output file, containing HEI-2010 component and total scores for each respondent are to be exported. Specify the name of the folder. */

filename RES “&home\RES”; /*In this Example, the folder is called “RES”, within the “home” folder, and the exported results will be a csv file called “hei2010r”. */
/*4. Read in required HEI-2010 scoring macros. These macros must be saved within the home folder. */

%include “&home\hei2010.beanspeas.allocation.macro.sas”;

%include “&home\hei2010.score.macro.sas”;

/*NOTE: Once you have completed all the steps above, all you need to do is run the SAS program below. Unless you used different names for your dataset and folders, no other action is required from you. */
TITLE 'HEI-2010 scores By Person for 2 days NHANES 2003-2004, AGE >= 2, RELIABLE DIETS, Include Pregnant and Lactating Women';

/*Step 1: locate the required datasets and variables and make necessary edits to the datasets*/
*part a: get MPED per 100 grams of food and perform edits and corrections;
data MPED;

 set MPED.EQUIV0304;

/*rename variables;  */
  FOODCODE=1*DRDIFDCD;

  MODCODE=1*DRDIMC; 

/*Replace some food codes with values to correct for previously identified errors in the database; */ 

*pizza (n=3) values from FPED 11/12 data;
if FOODCODE = 58106210 then do;

   G_TOTAL=1.88;

   G_WHL=0;

   G_NWHL=1.88;

   V_TOTAL=0.12;

   V_TOMATO=0.12;

   D_TOTAL=0.70;

   D_CHEESE=0.70;

   DISCFAT_OIL=0.44;

   DISCFAT_SOL=8.00;

   ADD_SUG=0.19;

   end;

else if FOODCODE = 58106220 then do;

   G_TOTAL=1.75;

   G_WHL=0;

   G_NWHL=1.75;

   V_TOTAL=0.12;

   V_TOMATO=0.12;

   D_TOTAL=0.66;

   D_CHEESE=0.66;

   DISCFAT_OIL=0.44;

   DISCFAT_SOL=10.62;

   ADD_SUG=0.19;

   end;

else if FOODCODE = 58106230 then do;

   G_TOTAL=1.88;

   G_WHL=0;

   G_NWHL=1.88;

   V_TOTAL=0.12;

   V_TOMATO=0.12;

   D_TOTAL=0.66;

   D_CHEESE=0.66;

   DISCFAT_OIL=0.44;

   DISCFAT_SOL=8.82;

   ADD_SUG=0.19;

   end;

/* Move soy beverages out of soybean products into dairy; */
/*In the HEI-2010, soy beverages are counted as part of the Milk component. Convert the four soy beverage codes in the MPED from M_SOY oz equivalents to D_TOTAL cup equivalents using the following conversion process*/
/*FOODCODE=11310000, MILK, IMITATION, FLUID, SOY BASED (1 cup=244 grams)

FOODCODE=11320000, MILK, SOY, READY-TO-DRINK, NOT BABY (1 cup=245 grams)

FOODCODE=11321000, MILK, SOY, READY-TO-DRINK, NOT BABY'S, CHOCOLATE (1 cup=240 grams)

FOODCODE=11330000, MILK, SOY, DRY, RECONSTITUTED, NOT BABY (1 cup=245 grams)*/
  if FOODCODE=11310000 then do;

     M_SOY=0;


 D_TOTAL=round(100*(1/244),.001);


 end;

  else if FOODCODE=11320000 then do;

     M_SOY=0;


 D_TOTAL=round (100*(1/245),.001);


 end;

  else if FOODCODE=11321000 then do;

     M_SOY=0;


 D_TOTAL=round (100*(1/240),.001);


 end;

  else if FOODCODE=11330000 then do;

     M_SOY=0;


 D_TOTAL=round (100*(1/245),.001);


 end;

run;

*part b: get whole fruit data per 100 grams of food;
data CNPPMPED_WJFRT;

  set MPED.cnppmyp_v1nhanes0304_wjfrt; 

run;

*part c: get individual food intake data for people with satisfactory recall status;
data FOOD1; 

  set NH.DR1IFF_C;

  FOODCODE=1*DR1IFDCD; /*convert variable name and type*/
  MODCODE=1*DR1MC;

  if DR1DRSTZ=1; /*reliable dietary recall status*/
  GRMS=DR1IGRMS;

  DAYREC=1;

run;

data FOOD2; 

  set NH.DR2IFF_C;

  FOODCODE=1*DR2IFDCD; /*convert variable name and type*/
  MODCODE=1*DR2MC;

  if DR2DRSTZ=1; /*reliable dietary recall status*/
  GRMS=DR2IGRMS;

  DAYREC=2;

run;

data FOOD (keep=SEQN FOODCODE MODCODE GRMS DAYREC);

  set FOOD1 FOOD2;

run;

*part d: get individual total food intake if reliable recall status;
data NUTRIENT1 (keep= SEQN WTDRD1 KCAL SFAT ALCO SODI DR1DRSTZ MFAT

   PFAT DAYREC);

  set NH.DR1TOT_C;

  if DR1DRSTZ=1; /*reliable dietary recall status*/
  KCAL=DR1TKCAL;

  MFAT=DR1TMFAT;

  PFAT=DR1TPFAT;

  SFAT=DR1TSFAT;

  SODI=DR1TSODI;

  ALCO=DR1TALCO;  

  DAYREC=1;

run;

data NUTRIENT2 (keep= SEQN KCAL SFAT ALCO SODI DR2DRSTZ MFAT

   PFAT DAYREC);

  set NH.DR2TOT_C;

  if DR2DRSTZ=1; /*reliable dietary recall status*/
  KCAL=DR2TKCAL;

  MFAT=DR2TMFAT;

  PFAT=DR2TPFAT;

  SFAT=DR2TSFAT;

  SODI=DR2TSODI;

  ALCO=DR2TALCO;

  DAYREC=2;  

run;

data NUTRIENT;

  set NUTRIENT1 NUTRIENT2;

run;

*part e: get demographic data for persons aged two and older;
data DEMO (keep=SEQN RIDAGEYR RIAGENDR SDDSRVYR SDMVPSU SDMVSTRA);

  set NH.DEMO_C;

  if RIDAGEYR >= 2;

run;

/*Step 2: Combine the required datasets*/
*part a: combine MPED and WHOLE FRUIT data on a food level;
proc sort data=MPED;

  by FOODCODE MODCODE;

run;

proc sort data=CNPPMPED_WJFRT;

  by FOODCODE MODCODE;

run;

data NEWMPED;

  merge MPED CNPPMPED_WJFRT;

  by FOODCODE MODCODE;

run;

*part b: combine food intake and MPED plus WHOLE FRUIT data on a food level;
proc sort data=FOOD;

  by FOODCODE MODCODE;

run;

data FDPYR;

  merge FOOD (in=S) NEWMPED (in=N);

  by FOODCODE MODCODE;

  if S and N;

run;

*part c: convert individuals' food intake amounts from grams to MyPyramid equivalents;
data FDPYR;

  set FDPYR;

 array PYRVAR G_TOTAL--A_BEV WHOLEFRT FRTJUICE;

 do over PYRVAR;

    PYRVAR=PYRVAR*(GRMS/100);

 end;

run;

*part d: calculate individual food intake amounts for MyPyramid food groups for each day;
proc sort data=FDPYR;

  by SEQN DAYREC;

run;

proc means data=FDPYR noprint;

  by SEQN DAYREC;

  var G_TOTAL--A_BEV WHOLEFRT FRTJUICE;

  output out=PYRCALC sum= ;

run;

*part e: combine nutrient and mped data on a person day level;
proc sort data=NUTRIENT;

  by SEQN DAYREC;

run;

data NUTFDPYR;

  merge NUTRIENT (in=N) PYRCALC;

  by SEQN DAYREC;

  if N;

run;

*part f: combine all data on a person day level;
proc sort data=DEMO;

  by SEQN;

run;

data BOTH;

  merge DEMO (in=C) NUTFDPYR (in=P);

  by SEQN;

  if C and P;   

run;

/*Please note: MyPyramid equivalent values for total vegetable intake (V_TOTAL) in the HEI-2010may be different from V_TOTAL in the MPED because legumes may be counted as vegetables or meat in the HEI-2010; and total dairy intake (D_TOTAL) in the HEI-2010 may be different from D_TOTAL in the MPED because soy beverages are counted as milk in the HEI-2010.*/
*part g: set pryvar to zero for people with zero cal who do not appear in iff data;
data BOTH;

  set BOTH;

  by SEQN;

 array PYRVAR G_TOTAL--A_BEV WHOLEFRT FRTJUICE;

 do over PYRVAR;

    if PYRVAR < 0 then PYRVAR =0;

 end;

run;

/*Step 3: Creates additional required variables: MONOPOLY, ALLMEAT, SEAPLANT, ADDSUGC, SOLFATC, EXALCCAL, and EMPTYCAL10. */
data BOTH;

  set BOTH;

  by SEQN;

  MONOPOLY=MFAT+PFAT;

  ALLMEAT=M_MPF+M_EGG+M_NUTSD+M_SOY;

  SEAPLANT=M_FISH_HI+M_FISH_LO+M_SOY+M_NUTSD;

 /**Calculate intake of Empty Calories**/
      ADDSUGC=16*ADD_SUG;
/*calories from added sugars*/
      SOLFATC=DISCFAT_SOL*9;
/*calories from solid fat*/
      MAXALCGR=13*(KCAL/1000);  /*max grams of alcohol based on kcal intake*/
      if ALCO <= MAXALCGR then EXALCCAL=0;  /*consumed less than max*/
        else if ALCO > MAXALCGR then EXALCCAL=7*(ALCO - MAXALCGR);  /*get cal from extra alc grams*/
      EMPTYCAL10=ADDSUGC+SOLFATC+EXALCCAL; /*total empty calories in hei2010 definition*/
run; 

*get sum per person of variables of interest;
proc means data=BOTH noprint;

  by SEQN;

  var KCAL ALLMEAT SEAPLANT V_TOTAL V_DRKGR LEGUMES F_TOTAL WHOLEFRT G_WHL D_TOTAL

    MONOPOLY SFAT SODI G_NWHL EMPTYCAL10;

  output out=BYID sum=;

run;

/*Step 4: Run the macro to properly allocate the intakes of the MPED variable Legumes (LEGUMES) to either Total Protein Foods and Seafood and Plant Proteins (ALLMEAT and SEAPLANT) or Total Vegetables and Greens and Beans (V_TOTAL and V_DRKGR).  The four resulting variables from this step, named LEGUME_ADDED_V_TOTAL, LEGUME_ADDED_BEANGRN, LEGUME_ADDED_ALLMEAT, and LEGUME_ADDED_SEAPLANT are all used in the HEI scoring macro in Step 5*/
%LEG2010A (indat=BYID,

           kcal= KCAL,

           allmeat=ALLMEAT,


   seaplant=SEAPLANT,


   v_total=V_TOTAL,


   v_drkgr=V_DRKGR,


   legumes=LEGUMES,


   outdat=AFTERLEG);

/*Step 5: Apply the HEI-2010 scoring macro. */
%HEI2010 (indat= AFTERLEG,

          kcal= KCAL,


  lv_total=LEGUME_ADDED_V_TOTAL,


  lbeangrn=LEGUME_ADDED_BEANGRN,


  f_total=F_TOTAL,


  wholefrt=WHOLEFRT,


  g_whl=G_WHL,


  d_total=D_TOTAL,

       lallmeat=LEGUME_ADDED_ALLMEAT,


  lseaplant=LEGUME_ADDED_SEAPLANT,


  monopoly=MONOPOLY,


  sfat=SFAT,


  sodi=SODI,


  g_nwhl=G_NWHL,


  Emptycal10=EMPTYCAL10,


  outdat=HEI2010);

/*Step 6: Displays and saves the results. */ 

*part a: this program saves one HEI-2010 score for each individual, based on one 24HR;
data HEI2010R (keep=SEQN KCAL HEIX1_TOTALVEG HEIX2_GREEN_AND_BEAN HEIX3_TOTALFRUIT

     HEIX4_WHOLEFRUIT HEIX5_WHOLEGRAIN HEIX6_TOTALDAIRY HEIX7_TOTPROT HEIX8_SEAPLANT_PROT

     HEIX9_FATTYACID HEIX10_SODIUM HEIX11_REFINEDGRAIN HEIX12_SOFAAS

     HEI2010_TOTAL_SCORE);

  set HEI2010;

  run;

*part b: calculates an unweighted mean across all individuals in group;
proc means n nmiss min max mean data=HEI2010R;

run;

*part c: saves results as CSV file;
proc export data=HEI2010R

  file=RES

  dbms=csv

  replace;

run;

